Artin L-functions

A tutorial

B. Allombert

IMB
CNRS/Université de Bordeaux

21/06/2017

Galois group

We start with a Galois extension of the rationale, here $\mathbb{Q}(\sqrt[3]{2}, \zeta_3) = \mathbb{Q}(\sqrt[6]{-108})$, with Galois group isomorphic to $\mathfrak{S}_3$.

```
? N = nfinit(x^6+108);
? G = galoisinit(N);
```

G is the Galois group of N.

Linear representation

```
? [T,o] = galoischartable(G);
? T~
%4 = [1  1  1]
%    [1  1 -1]
%    [2 -1  0]
```

T is the character table of $G \cong \mathfrak{S}_3$, which is defined over $\mathbb{Z}$.
The first character is related to the trivial representation, the second to the signature, and the third to a faithful irreducible representation of dimension 2.
The ordering of the conjugacy classes is given by `galoisconjclasses(G)`.
We will compute the Artin function associated to the third character.

Artin L-function

```
? L = lfunartin(N,G,T[,3],o);
? lfuncheckfeq(L)
%6 = -127
? L[2..5]
%7 = [0, [1], 1, 108]
? z = lfun(L,0,1)
%8 = 1.3473773483293841009181878914456530463
? p = algdep(exp(z),3)
%9 = x^3-3*x^2-3*x-1
```

which suggests that this function is equal to a Hecke L-function.

```
? bnr = bnrinit(bnfinit(a^2+a+1),6);
? lfunan([bnr,[1]],100)==lfunan(L,100)
%11 = 1
```

A more interesting example

Let E be a model of the curve $X_0(11)$
$E : y^2 + y = x^3 - x^2 - 10x - 20$, we build the field $\mathbb{Q}(E[3])$
generated by the coordinates of the points of 3-torsions.

```
? default(parisize,"16M");
? E=ellinit([0,-1,1,-10,-20]);
  \\ or ellinit("11a1") if elldata is available
? P=elldivpol(E,3)
%13 = 3*x^4-4*x^3-60*x^2-237*x-21
? Q=polresultant(P,y^2-elldivpol(E,2));
%14 = 27*y^8+108*y^7-4813*y^6-14817*y^5+162543*y^4+
? R=nfsplitting(Q)
%15 = y^48-36*y^46+558*y^44-4588*y^42+24549*y^40-11
```

This defines a Galois extension of $\mathbb{Q}$ with Galois group $\mathrm{GL}_2(\mathbb{F}_3)$.

Non monomial representation

```
? N=nfinit(R); G=galoisinit(N);
? [T,o]=galoischartable(G); T~
%18 = [1,1,1,1,1,1,1,1;
%      1,1,-1,1,1,1,-1,-1;
%      2,0,y^5+y^3-y,1,-1,-2,0,-y^5-y^3+y;
%      2,0,-y^5-y^3+y,1,-1,-2,0,y^5+y^3-y;
%      2,2,0,-1,-1,2,0,0;
%      3,-1,-1,0,0,3,1,-1;
%      3,-1,1,0,0,3,-1,1;
%      4,0,0,-1,1,-4,0,0]
? o
%19 = 24
```

$o = 24$ means that the variable y denotes a 24-th root of unity.

Non monomial representation

```
? minpoly(Mod(y^5+y^3-y, polcyclo(24,y)))
%21 = x^2+2
```

So the coefficients are in $\mathbb{Q}(\sqrt{-2})$. We use the third irreducible representation.

```
? L = lfunartin(N,G,T[,3],o);
? L[2..5]
%23 = [0,[0,1],1,3267]
? lfuncheckfeq(L)
%24 = -127
```

Determinant

```
? dT = galoischardet(G,T[,3],o)
%25 = [1,-1,-1,1,1,1,1,-1]~
? dL = lfunartin(N,G,dT,o);
? dL[2..5]
%27 = [0,[1],1,3];
```

So L is associate to a modular form of weight 1, level 3267 and Nebentypus $\left(\frac{-3}{.}\right)$.

Link to E

We reduce the coefficients of L modulo $1+\sqrt{-2}$ of norm 3.

```
? S = lfunan(L,1000); SE = lfunan(E,1000);
? Smod3 = round(real(S))-round(imag(S)/sqrt(2));
? [(Smod3[i]-SE[i])%3|i<-[1..#Smod3],gcd(i,33)==1]
%29 = [0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,...
```

The coefficients of L are congruent to the coefficients of the L-function associated to E modulo $1+\sqrt{-2}$.

Quotient of Hecke L-function

We will write L as L_1/L_2, where L_1 and L_2 are two Hecke L-functions.

```
? bnf6=bnfinit(a^6-3*a^5+6*a^4+4*a^3+6*a^2-3*a+1);
? bnr6=bnrinit(bnf6,1);
? bnf4=bnfinit(a^4-a^3+3*a^2+a-1);
? pr4 = idealprimedec(bnf4,3)[1];
? bnr4=bnrinit(bnf4,[pr4,[0,1]]);
? L1=lfuncreate([bnr6,[5]]);
? L1[2..5]
%36 = [1, [0, 0, 0, 1, 1, 1], 1, 32019867]
? L2=lfuncreate([bnr4,[1]]);
? L2[2..5]
%38 = [0, [0, 0, 1, 1], 1, 9801]
```

Quotient of Hecke L-function

```
? LL = lfundiv(L1,L2);
? round(lfunan(L,1000)-lfunan(LL,1000),&e)
%40 = [0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,...
? e
%41 = -125
```

So L is equal to a quotient of two Hecke L-functions.